
FRACTAL

PURPOSE
Generate a fractal sequence from a set of points.

DESCRIPTION
Given as set of points in the variables X1 and Y1, this command generates the following sequence:

X2(1) = X1(1)
X2(2) = X1(1) + (1/3)DELX
X2(3) = X1(1) + (1/2)DEXLX - (SQRT(3)/6)DELY
X2(4) = X1(1) + (2/3)DELX
X2(5) = X1(2)
Y2(1) = Y1(1)
Y2(2) = Y1(1) + (1/3)DELY
Y2(3) = Y1(1) + (1/2)DELY - (SQRT(3)/6)DELX
Y2(4) = Y1(1) + (2/3)DELY
Y2(5) = Y1(2)

where DELX is the difference between 2 successive points in the X1 variable (i.e., DELX(I) = X1(I+1)-X1(I)) and DELY is the
difference between 2 successive points in the Y1 variable (i.e., DELY(I) = Y1(I+1)-Y1(I)). This is the type of fractal that can be used to
generate a Koch snowflake or a Koch curve.

SYNTAX
LET <x2> <y2> = FRACTAL <x1> <y1> <SUBSET/EXCEPT/FOR qualification>
where <x1> is a variable;

 <y1> is a variable (same length as <x1>);
 <x2> is a variable where the computed fractal sequence corresponding to <x1> is saved (4 times as long as <x1>);
 <y2> is a variable where the computed fractal sequence corresponding to <y1> is saved (4 times as long as <y1>);

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

EXAMPLES
LET Y2 X2 = FRACTAL Y1 X1

DEFAULT
None

SYNONYMS
None

RELATED COMMANDS
FRACTAL PLOT = Generate an iterated function system type fractal plot.
CANTOR NUMBERS = Generate a sequence of Cantor numbers.
LOGISTIC NUMBERS = Generate numbers from the logistic sequence.
JULIA = Compute a Julia set.
PHASE PLANE DIAGRAM = Generate a phase plane diagram.
PLOT = Plots data or functions.

REFERENCE
“Infinity and the Mind,” Rucker, (page 9).

APPLICATIONS
Chaos

IMPLEMENTATION DATE
88/7

PROGRAM 1
READ X1 Y1
0 1
0.5 0
1 1
0 1
END OF DATA
.
MULTIPLOT 2 3
MULTIPLOT CORNER COORDINATES 0 0 100 100
LIMITS 0 1
TIC OFFSET 0.4 0.4
FRAME OFF
FRAME CORNER COORDINATES 5 5 95 95
TITLE OFFSET 1
PRE-SORT OFF
TITLE ORIGINAL DATA
PLOT Y1 X1
LOOP FOR K = 2 1 6
    LET J = K - 1
    LET Y2 X2 = FRACTAL Y1 X1
    TITLE KOCH SNOWFLAKE ITERATION ^K
    PLOT Y2 X2
    LET X1 = X2
    LET Y1 = Y2
END OF LOOP
END OF MULTIPLOT

ORIGINAL DATA KOCH SNOWFLAKE ITERATION 2 KOCH SNOWFLAKE ITERATION 3

KOCH SNOWFLAKE ITERATION 4 KOCH SNOWFLAKE ITERATION 5 KOCH SNOWFLAKE ITERATION 6

PROGRAM 2
.  Draw a Koch curve through several iterations
READ X1 Y1
0 0
1 0
END OF DATA
MULTIPLOT 2 3
MULTIPLOT CORNER COORDINATES 0 0 100 100
LIMITS 0 1
TIC OFFSET 0.4 0.4
FRAME OFF
FRAME CORNER COORDINATES 5 5 95 95
TITLE OFFSET 1
PRE-SORT OFF
TITLE ORIGINAL LINE
PLOT Y1 X1
LOOP FOR K = 2 1 6

LET J = K - 1
LET Y2 X2 = FRACTAL Y1 X1
TITLE KOCH CURVE ITERATION ^K
PLOT Y2 X2
LET X1 = X2
LET Y1 = Y2

END OF LOOP
END OF MULTIPLOT

ORIGINAL LINE KOCH CURVE ITERATION 2 KOCH CURVE ITERATION 3

KOCH CURVE ITERATION 4 KOCH CURVE ITERATION 5 KOCH CURVE ITERATION 6


